NAG Library Routine Document

G05SEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05SEF generates a vector of pseudorandom numbers taken from a Dirichlet distribution.

2 Specification

```
SUBROUTINE G05SEF (N, M, A, STATE, X, LDX, IFAIL)

INTEGER            N, M, STATE(*), LDX, IFAIL
REAL (KIND=nag_wp) A(M), X(LDX,M)
```

3 Description

The distribution has PDF (probability density function)

$$f(x) = \frac{1}{B(\alpha)}\prod_{i=1}^{m} x_i^{\alpha_i - 1} \quad \text{and}$$

$$B(\alpha) = \frac{\prod_{i=1}^{m}\Gamma(\alpha_i)}{\Gamma\left(\sum_{i=1}^{m}\alpha_i\right)}$$

where $x = \{x_1, x_2, \ldots, x_m\}$ is a vector of dimension m, such that $x_i > 0$ for all i and $\sum_{i=1}^{m} x_i = 1$.

G05SEF generates a draw from a Dirichlet distribution by first drawing m independent samples, $y_i \sim \text{gamma}(\alpha_i, 1)$, i.e., independent draws from a gamma distribution with parameters $\alpha_i > 0$ and one, and then setting $x_i = y_i / \sum_{j=1}^{m} y_j$.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05SEF.

4 References

Dagpunar J (1988) *Principles of Random Variate Generation* Oxford University Press

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of pseudorandom numbers to be generated.

Constraint: N ≥ 0.

2: M – INTEGER *Input*

On entry: m, the number of dimensions of the distribution.

Constraint: M > 0.

3: A(M) – REAL (KIND=nag_wp) array *Input*

On entry: the parameter vector for the distribution.

Constraint: A$(i) > 0.0$, for $i = 1, 2, \ldots,$ M.

4: STATE$(*)$ – INTEGER array *Communication Array*

Note: the actual argument supplied must be the array STATE supplied to the initialization routines G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

5: X(LDX,M) – REAL (KIND=nag_wp) array *Output*

On exit: the n pseudorandom numbers from the specified Dirichlet distribution, with X(i, j) holding the jth dimension for the ith variate.

6: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G05SEF is called.

Constraint: LDX $\geq$ N.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N < 0.

IFAIL $= 2$

On entry, M < 1.

IFAIL $= 3$

On entry, at least one A$(i) \leq 0.0$.

IFAIL = 4

On entry, STATE vector was not initialized or has been corrupted.

IFAIL = 6

On entry, LDX < N.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints a set of five pseudorandom numbers from a Dirichlet distribution with parameters $m = 4$ and $\alpha = \{2.0, 2.0, 2.0, 2.0\}$, generated by a single call to G05SEF, after initialization by G05KFF.

9.1 Program Text

```
    Program g05sefe

!     G05SEF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g05kff, g05sef, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: lseed = 1, nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: genid, ifail, ldx, lstate, m, n, subid
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:), x(:,:)
      Integer                          :: seed(lseed)
      Integer, Allocatable             :: state(:)
!     .. Executable Statements ..
      Write (nout,*) 'G05SEF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!     Initial call to initialiser to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!     Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)
```

```
!     Read in sample size and number of dimensions
      Read (nin,*) n, m

      ldx = n
      Allocate (x(ldx,m),a(m))

!     Read in the distribution parameters
      Read (nin,*) a(1:m)

!     Generate the variates
      ifail = 0
      Call g05sef(n,m,a,state,x,ldx,ifail)

!     Display the variates
      ifail = 0
      Call x04caf('General',' ',n,m,x,ldx,' ',ifail)

    End Program g05sefe
```

9.2 Program Data

```
G05SEF Example Program Data
1  1  1762543     :: GENID,SUBID,SEED(1)
5 4               :: N,M
2.0 2.0 2.0 2.0   :: A
```

9.3 Program Results

```
 G05SEF Example Program Results

          1       2       3       4
 1   0.3600  0.3138  0.0837  0.2426
 2   0.2874  0.5121  0.1497  0.0509
 3   0.2286  0.2190  0.3959  0.1566
 4   0.1744  0.3961  0.2764  0.1530
 5   0.1522  0.2845  0.2074  0.3559
```